

SQL ci parla *diagnostics area*



Marco Riva




www.markonetools.it



Ultimo aggiornamento: 29/04/2022

Come ci parla l'SQL?



SQLCA – Communication Area



- ▶ E' la **struttura dati** fondamentale con cui SQL ci restituisce informazioni sull'esito di qualsiasi istruzione SQL
- ▶ In RPG è definita dal precompilatore SQL
- ▶ Viene valorizzata dopo ogni istruzione SQL
- ▶ E' da utilizzare *obbligatoriamente* per conoscere l'esito dell'istruzione (tranne istruzioni `DECLARE`). Embedded SQL non restituirà mai un'eccezione RPG come con I/O nativo
- ▶ In ogni programma esiste **una sola** SQLCA



3

SQL Diagnostics Area



- ▶ E' un insieme di informazioni aggiuntive rispetto a quanto restituito in SQLCA
- ▶ Non restituisce informazioni solo in caso di errore
- ▶ Vengono reperite tramite l'istruzione SQL get diagnostics



4

SQLDA – SQL Descriptor Area



- E' un insieme di variabili utilizzate nell'istruzione `describe` oppure dalle istruzioni `prepare`, `open`, `call`, `fetch` e `execute`
- In RPG l'SQLDA è definita dal precompilatore SQL
- L'SQLDA utilizzata in `describe` o `prepare` restituisce informazioni sull'istruzione `select` che è stata preparata



5


Get diagnostics

- ▶ L'istruzione SQL **get diagnostics** restituisce varie informazioni sull'istruzione SQL eseguita immediatamente prima
- ▶ Non può essere preparata dinamicamente

ESEMPIO

```
exec sql
  open curs01;

if sqlcode < 0;
exec sql
  get diagnostics condition 1
    :MsgErrore = MESSAGE_TEXT,
    :MsgID = DB2_MESSAGE_ID,
    :MsgIDCPF = DB2_MESSAGE_ID1,
    :MsgIDCPD = DB2_MESSAGE_ID2,
    :MsgLength = MESSAGE_LENGTH,
    :db2sqlcode = DB2_RETURNED_SQLCODE,
    :db2sqlstate = RETURNED_SQLSTATE;
```



sqlcode e sqlstate
dell'istruzione SQL
precedente a get
diagnostics

Tipi di informazioni

- ▶ *statement-information*: restituisce informazioni sull'ultima istruzione SQL
- ▶ *condition-information*: restituisce informazioni sulle condizioni verificatesi durante l'esecuzione dell'ultima istruzione SQL. La condizione **1** si riferisce alla condizione indicata da SQLSTATE
- ▶ *combined-information*: restituisce le informazioni statement o condition combinate in una sola stringa



7

SQLCODE / SQLSTATE

ATTENZIONE

- ▶ l'esecuzione di `get diagnostics` modifica `SQLCODE` e `SQLSTATE`
- ▶ Per avere i valori originali di `sqlcode` e `sqlstate` prima dell'esecuzione di `get diagnostics` usare i parametri `DB2_RETURNED_SQLCODE` e `RETURNED_SQLSTATE`



Statement-information

- ▶ `COMMAND_FUNCTION`, `COMMAND_FUNCTION_CODE`: nome e codice dell'istruzione eseguita
- ▶ `DYNAMIC_FUNCTION`, `DYNAMIC_FUNCTION_CODE`: nome e codice dell'istruzione dinamicamente preparata
- ▶ `DB2_LAST_ROW`: per fetch a blocchi contiene +100 se si è raggiunto l'ultimo record
- ▶ `DB2_NUMBER_PARAMETER_MARKERS`: numero dei parameter markers (?) presenti nell'istruzione prepare
- ▶ `DB2_NUMBER_ROWS`: per open o fetch può restituire il numero di righe (~stima) del resultset
- ▶ `ROW_COUNT`: per delete, insert, refresh, update, merge, fetch a blocchi restituisce il numero di righe interessate dall'istruzione
- ▶ `DB2_SQL_ATTR_CURSOR_CAPABILITY`: per open indica la capacità del cursore (R read-only, D read/delete, U read/delete/update)



Statement-information: esempio/1

```
dcl-s RowNumber packed(31);
dcl-s CmdFunc varchar(128);
dcl-s CmdFuncCode int(10);
dcl-s CurType char(1);

exec sql
    open CURS05;
exec sql
    get diagnostics :RowNumber = DB2_NUMBER_ROWS,
                   :CmdFunc = COMMAND_FUNCTION,
                   :CmdFuncCode = COMMAND_FUNCTION_CODE,
                   :CurType = DB2_SQL_ATTR_CURSOR_CAPABILITY;
```



Statement-information: esempio/2

```
dcl-s RowNumber packed(31);  
exec sql  
    insert into EMPLOYEE  
        select * from MY_EMP_TMP where EMPNO >= '500000';  
exec sql  
    get diagnostics :RowNumber = ROW_COUNT;  
// RowNumber = numero record aggiunti
```



Condition-information

- ▶ `COLUMN_NAME`: colonna che ha causato l'errore con `SQLSTATE` = classe 42
- ▶ `CONSTRAINT_NAME`, `CONSTRAINT_SCHEMA`: libreria e nome del vincolo violato quando `SQLSTATE` = classe 23 o 27 o 40002
- ▶ `DB2_MESSAGE_ID`: id del messaggio corrispondente a `MESSAGE_TEXT`
- ▶ `DB2_MESSAGE_ID1`: id messaggio CPF
- ▶ `DB2_MESSAGE_ID2`: id messaggio CPD
- ▶ `MESSAGE_TEXT`, `MESSAGE_LENGTH`: testo e lunghezza del messaggio
- ▶ `DB2_RETURNED_SQLCODE`, `RETURNED_SQLSTATE`: `SQLCODE` e `SQLSTATE` dell'istruzione che si sta diagnosticando
- ▶ `DB2_ROW_NUMBER`: per insert o fetch a blocchi restituisce il numero di righe
- ▶ `DB2_SQLERRDn`: corrisponde a `SQLERRD (n)` presente in `SQLCA`



Condition-information: esempio

```
dcl-s MsgErrore varchar(32740);
dcl-s MsgID char(10);
dcl-s MsgIDCPF varchar(7);
dcl-s MsgIDCPD varchar(7);
dcl-s MsgLength int(10);
dcl-s db2SqlCode int(10);
dcl-s db2SqlState char(5);

exec sql
  open CURS05;
if sqlcod < 0;
  exec sql
    get diagnostics condition 1 :MsgErrore = MESSAGE_TEXT,
                                :MsgID = DB2_MESSAGE_ID,
                                :MsgIDCPF = DB2_MESSAGE_ID1,
                                :MsgIDCPD = DB2_MESSAGE_ID2,
                                :MsgLength = MESSAGE_LENGTH,
                                :db2sqlcode = DB2_RETURNED_SQLCODE,
                                :db2sqlstate = RETURNED_SQLSTATE;
endif;
```



Combined-information

► Esempio **all condition**

```
CLASS_ORIGIN=ISO
9075;CONDITION_NUMBER=+1;CURSOR_NAME=CURS05;DB2_MESSAGE_ID=SQL0502;DB2_MODULE_DETECTING_ERROR=QSQOPEN;DB2_ORDINAL_TOKEN_1=CURS05;DB2_RETURNED_SQLCODE=-
502;DB2_TOKEN_COUNT=+1;DB2_TOKEN_STRING=CURS05;MESSAGE_LENGTH=+42;MESSAGE_OCTET_LENGTH=+42;MESSAGE_TEXT=Il cursore CURS05 è già aperto o allocato.;RETURNED_SQLSTATE=24502;SERVER_NAME=SIRIODEMOMI;DB2_SQLERRD_SET=Y;
```

► Esempio **all statement**

```
COMMAND_FUNCTION=OPEN;COMMAND_FUNCTION_CODE=+53;DB2_NUMBER_CONNECTIONS=+1;DB2_NUMBER_ROWS=+5;DB2_ROW_LENGTH=+104;DB2_SQL_ATTR_CONCURRENCY=R;DB2_SQL_ATTR_CURSOR_CAPABILITY=R;DB2_SQL_ATTR_CURSOR_HOLD=N;DB2_SQL_ATTR_CURSOR_ROWSET=N;DB2_SQL_ATTR_CURSOR_SCROLLABLE=N;DB2_SQL_ATTR_CURSOR_SENSITIVITY=S;DB2_SQL_ATTR_CURSOR_TYPE=F;MORE=N;NUMBER=+1;
```



Combined-information: esempio

```
dcl-s TxtDiagno varchar(500);  
exec sql  
    open CURS05;  
  
if sqlcod < 0;  
    exec sql  
        get diagnostics :TxtDiagno = all condition;  
else;  
    exec sql  
        get diagnostics :TxtDiagno = all statement;  
endif;
```



Riferimenti



➤ E-mail aziendale: mriva@sirio-is.it



➤ Blog: www.markonetools.it



➤ E-mail blog: info@markonetools.it



➤ LinkedIn: www.linkedin.com/in/marcoriva-mk1



➤ Twitter: [@MarcoRiva73](https://twitter.com/MarcoRiva73)



➤ Facebook: <https://www.facebook.com/markonetools/>



➤ YouTube: <https://www.youtube.com/channel/UCb47YJQJCzU-5x4nnGzDu-w>

Power coffee - MK1



16